Добрый день!

Написал простой запрос:

```1c
    ВЫБРАТЬ
    СчетКлиенту.Ссылка КАК СчетКлиенту,
    Отчет.Ссылка КАК Отчет
    ПОМЕСТИТЬ Счета
    ИЗ
    Документ.СчетКлиенту КАК СчетКлиенту
    ЛЕВОЕ СОЕДИНЕНИЕ Документ.Отчет КАК Отчет
    ПО (Отчет.Основание = СчетКлиенту.Ссылка)
    ГДЕ
    НЕ Отчет.Основание ЕСТЬ NULL
    И НЕ Отчет.ПометкаУдаления
    И СчетКлиенту.Дата МЕЖДУ &Дата1 И &Дата2
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ РАЗЛИЧНЫЕ
    Счета.СчетКлиенту КАК СчетКлиенту,
    Счета.СчетКлиенту.НомерСчета КАК НомерСчета,
    Счета.СчетКлиенту.Контрагент КАК Контрагент,
    Счета.ОтчетЭкспедитора КАК ОтчетЭкспедитора
    ИЗ
    Счета КАК Счета,
    Документ.РеестрДокументов.ДокументыРеестра КАК РеестрДокументовДокументыРеестра
    ГДЕ
    НЕ Счет.СчетКлиенту В
    (ВЫБРАТЬ
    РеестрДокументовДокументыРеестра.Счет
    ИЗ
    Документ.РеестрДокументов.ДокументыРеестра КАК РеестрДокументовДокументыРеестра)
```

Задача была получить запросом все документы "Счет", на основании которых введен документ "Отчет", и, которые отсутствуют в ТЧ документов "Реестр документов".
Запрос рабочий, но выполняется жуть как долго. Документов в базе много. Итоговый запрос с отбором за месяц выполняется 1 минуту 20 секунд!! Результат содержит 1000 строк.
Изначально в планах не предполагалось отбора по дате, но без этого даже первый подзапрос отрабатывает минутами.
Подскажите, пожалуйста, какие есть варианты ускорить выполнение запроса? Возможно переделать какие-то конструкции? Заранее спасибо.
---
Поля "НомерСчета", "Контрагент", "ОтчетЭкспедитора" нужно выбирать сразу во временную таблицу "Счета".
Условие "В (Выбрать ... )" заменить на соединение таблиц.
---
Счета КАК Счета,
    Документ.РеестрДокументов.ДокументыРеестра КАК РеестрДокументовДокументыРеестра

Вы не указали по каким полям связывать 2 таблицы.
По логике запроса строка Документ.РеестрДокументов.ДокументыРеестра КАК РеестрДокументовДокументыРеестра вообще не нужна.
---
Поля "НомерСчета", "Контрагент", "ОтчетЭкспедитора" нужно выбирать сразу во временную таблицу "Счета".
Условие "В (Выбрать ... )" заменить на соединение таблиц.
---
Всем спасибо за комментарии. С учетом Ваших замечаний сделал так:

```1c
    ВЫБРАТЬ
    СчетКлиенту.Ссылка КАК СчетКлиенту,
    СчетКлиенту.Контрагент КАК Контрагент,
    СчетКлиенту.НомерСчета КАК НомерСчета,
    Отчет.Ссылка КАК Ссылка
    ПОМЕСТИТЬ ВременнаяТаблица
    ИЗ
    Документ.СчетКлиенту КАК СчетКлиенту
    ЛЕВОЕ СОЕДИНЕНИЕ Документ.Отчет КАК Отчет
    ПО (Отчет.Основание = СчетКлиенту.Ссылка)
    ГДЕ
    НЕ Отчет.Основание ЕСТЬ NULL
    И НЕ Отчет.ПометкаУдаления
    И СчетКлиенту.Дата МЕЖДУ &Дата1 И &Дата2
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ РАЗЛИЧНЫЕ
    ВременнаяТаблица.СчетКлиенту КАК СчетКлиенту,
    ВременнаяТаблица.Контрагент КАК Контрагент,
    ВременнаяТаблица.НомерСчета КАК НомерСчета,
    ВременнаяТаблица.Ссылка КАК Ссылка
    ИЗ
    ВременнаяТаблица КАК ВременнаяТаблица
    ЛЕВОЕ СОЕДИНЕНИЕ Документ.РеестрДокументов.ДокументыРеестра КАК РеестрДокументовДокументыРеестра
    ПО (ВременнаяТаблица.СчетКлиенту = РеестрДокументовДокументыРеестра.Счет)
    ГДЕ
    РеестрДокументовДокументыРеестра.Ссылка ЕСТЬ NULL
```

Время выполнения снизилось до ~2 секунд.

Всем еще раз спасибо
---
Я не понимаю этой строчки
    Счета КАК Счета,
  Документ.РеестрДокументов.ДокументыРеестра КАК РеестрДокументовДокументыРеестра
Для чего вы тянете ВСЮ таблицу в этом запросе?
---
А почему Вы используете "Левое соединение" в первом запросе? Думаю тут напрашивается "Внутренние". Вам интересны только те документы которые есть в обоих таблицах. И отборы я бы перенес в условия связи